Title

THE IBM SHARE PROGRAM D2 NU SCHR 1072 FOR SOLUTION OF THE SCHRODINGER RADIAL EQUATION, BY J.W. COOLEY: NECESSARY AND USEFUL MODIFICATIONS FOR ITS USE ON IBM 7090

Authors

Zare, R.N.
Cashion, J.K.

Publication Date

1963-07-01

UCRL-10881

University of California

Ernest O. Lawrence Radiation Laboratory

THE IBM SHARE PROGRAM D2 NU SCHR 1072 FOR SOLUTION OF THE SCHRODINGER RADIAL EQUATION, BY J.W. COOLEY: NECESSARY AND USEFUL MODIFICATIONS FOR ITS USE ON AN IBM 7090.

Berkeley, California

UCRL-10881

UNIVERSITY OF CALIFORNIA

Lawrence Radiation Laboratory
Berkeley, California

Contract No. W-7405-eng-48

THE IBM SHARE PROGRAM D2 NU SCHR 1072 FOR SOLUTION OF THE SCHRÖDINGER RADIAL EQUATION, BY J.W. COOLEY: NECESSARY AND USEFUL MODIFICATIONS FOR ITS USE ON AN IBM 7090

R. N. Zare and J. K. Cashion

July 1963

UNIVERSITY OF CALIFORNIA
Lawrence Radiation Laboratory
Berkeley, California

AEC Contract No. W-7405-eng-48

December 13, 1963

ERRATA

TO: All recipients of UCRL-10881

FROM: Technical Information Division

Subject: UCRL-10881, "The IBM SHARE Program D2 NU SCHR 1072 for Solution of the Schrodinger Radial Equation, by J. W. Cooley; Necessary and Useful Modifications for Its Use on an IBM 7090," R. N. Zare and J. K. Cashion, July 1963.

Please make the following corrections on subject report.

Page 18

1. Card "MOD 5" should be changed to read

```
899    FORMAT(36H DIFFERENCE EQUATION TECHNIQUE FAILS)    MOD 5
```

2. Insert between cards "MOD 5 and MOD 6" the following card

```
       SCHR = 1.                                          MOD 5.5
```

0 0 1 0 1 8 2 9 5 2 6

UCRL-10881

The IBM SHARE Program D2 NU SCHR 1072 for Solution of the Schrodinger Radial Equation, by J. W. Cooley; Necessary and Useful Modifications for Its Use on an IBM 7090.

R. N. Zare and J. K. Cashion

Department of Chemistry and Lawrence Radiation Laboratory, University of California, Berkeley, California

Introduction

An efficient method for solving the radial Schrodinger equation numerically has been proposed recently by Cooley.[1] He has contributed a Fortran program embodying his procedure to the IBM SHARE project, D2 NU SCHR 1072. The authors of the present report have found his program extremely useful as a means of generating eigenvalues and eigenfunctions for diatomic molecule calculations. The program was written for an IBM 704 and must be modified slightly in order to be compatible with an IBM 7090. (The same modifications are necessary for its use on the 709 or 7094.) The principal purpose of this report is to make this information available to 7090 users.

A few modifications and additions which the authors have found useful also will be described. These in no way derogate from Cooley's skill as a programmer. On the contrary his program can be recommended not only for its utility but also as an excellent example of the efficient use of Fortran coupled with a clarity of presentation which makes it easy for others to adapt his work to their own particular needs.

Tests of the accuracy of this procedure when up to 200 intervals are used in the integration have been given by Cooley.[1] Additional information on its use with up to 2000 intervals has been given elsewhere by one of (J.K.C.[2]). Other applications in which this program has been very useful to us include a study of vibration-rotation interaction (J.K.C.[3]) and a calculation of the intensity distribution in the iodine fluorescence spectrum based on Rydberg-Klein-Rees potentials (R.N.Z.[4]).

The Numerical Method

The following is taken largely from Cooley's summary of the method as given in the SHARE distribution description of this program. To facilitate comparison with the listing of the symbolic deck given in Appendix B, variable names employed in the program are also used in the following equations.

The program calculates the eigenvalue E and the normalized eigenfunction S of the second-order differential equation

$$\frac{d^2S}{dr^2} = (V - E)S \tag{1}$$

where V is a given numerical potential function.

It is customary to separate the Schrodinger equation for a diatomic molecule into its radial and angular parts and express its solution in the form

$$\psi(r,\theta,\varphi) = R(r)\ \Theta(\theta)\ \Phi(\varphi). \tag{2}$$

Eq. (1) is equivalent to the radial Schrodinger equation when length and energy are expressed in dimensionless units. Its solution S(r) is equal to r times R(r). S(r) provides a more convenient computational form since the product of any two functions S will include the volume element of integration. Hence, the expectation value for any quantity F is given by

$$[\overline{F}] = \int RF_{op}R \cdot r^2 dr = \int SF_{op}S\ dr \tag{3}$$

If length is measured in units of Bohr radii, $a_o = 0.529172$ Å, the energies V and E must be expressed in a dimensionless unit equivalent to $h\,N_o/8\pi^2 c a_o^2 \mu_A$ wave numbers, where N_o is Avagadro's number (physical scale) and μ_A is the reduced mass in Aston units. The numerical value of this factor is $60.2198/\mu_A$. Hence to convert eigenvalues generated by this program to the more familiar units of cm^{-1}, multiply them by this factor. Note that the Aston unit of reduced mass is based on the physical scale of atomic weight, i.e., $\mu_A = m_1 m_2/(m_1+m_2)$ where m_1 and m_2 are the atomic weights of atoms relative to $O^{16} = 16$. Herzberg's[5] table 39 is a convenient source for μ_A values.

Regarding the potential V(r) two things should be noted. First, its zero is taken at the dissociation limit. Therefore a Morse potential, for example, must be generated from the relation

$$V(r) = D_e[1 - e^{-\beta(r-r_e)}]^2 - D_e \qquad (4)$$

where the dissociation energy D_e is a positive quantity. Secondly, V(r) is an effective potential which may include a rotational term, $[J(J+1) - \Lambda]/r^2$, where J is the rotational quantum number and Λ is the quantum number for the z-component of electronic angular momentum. Since this term is inherently positive its addition to the potential for the rotationless state always raises the potential.

Initially, a non-normalized solution P(r) for Eq. (1) is calculated from the integration formula

$$-Y_{i-1} + 2Y_i - Y_{i+1} + h^2(V_i - E)P_i = 0 \qquad (5)$$

where

$$P_i = P(r_i)$$

$$Y_i = [1-(h^2/12)(V_i-E)]P_i \qquad (6)$$

$$V_i = V(r_i)$$

$$h = r_{i+1} - r_i.$$

The error associated with the use of Eq. (5) as a predictor is $(h^6/240)P_i^{(vi)}$.

Starting with the boundary values

$$P_n = 10^{-30} \qquad (7)$$

$$P_{n-1} = P_n \cdot \exp[r_n(V_n-E)^{1/2} - r_{n-1}(V_{n-1}-E)^{1/2}] \qquad (8)$$

and a trial value of E, Eq. (5) is used to integrate inward, giving P_{n-2}, P_{n-3}, ..., P_m, where m is selected as the first point for which $|P_m| \leq |P_{m+1}|$. [Eq. (8) is based on the WKB approximation.] Then, P_i is replaced by P_i/P_m for i = m, m+1, ..., n.

Starting with the boundary values

$$P_o = 0, \; P_1 = 10^{-20}$$

Eq. (5) is used to integrate outwards, giving P_2, ..., P_m, after which P_i is replaced by P_i/P_m for $i = 1,2,\cdots,m$. This yields a trial solution satisfying Eq. (5) at all points except r_m, the crossing-point for the outward and inward curves. A correction to E is calculated by applying the Newton-Raphson method to the calculation of the zeros of the function

$$F(E) = h^{-2}(-Y_{m-1} + 2Y_m - Y_{m+1}) + (V_m - E)P_m \tag{9}$$

which is a measure of the amount by which the m^{th} equation of Eq. (2) is not satisfied. The derivative of Eq. (9) is

$$F'(E) = -\sum_{i=1}^{n} P_i^2 \tag{10}$$

The correction to E, by the Newton-Raphson method, is

$$DE = -F(E)/F'(E). \tag{11}$$

After adding this correction to E to obtain a new trial eigenvalue the process is repeated until $DE \leqslant \epsilon$. When convergence is achieved, the normalized solution is calculated,

$$S_i = P_i \Big/ \left(h \sum_{j=1}^{n} P_j^2\right)^{1/2}, \quad i=1,2,\cdots,n. \tag{12}$$

Necessary Modifications

The SHARE distribution includes the main program NU SCHR either in the form of 177 Fortran source program cards or as 43 relocatable binary cards. In addition binary cards labelled NU EMFT, 1-4 and NU LRT, 1-6 are also supplied. For 7090 use, discard all ten of these binary cards. Replace them with the FAP deck for subroutine EMFT which is listed in Appendix A. When compiled this subroutine will be contained on four binary cards.

Useful Modifications

Appendix B contains a listing of one Fortran source deck used by the authors. Frequently a particular application will require recompilation with an altered DIMENSION statement, probably the inclusion of a COMMON card and perhaps some modification of output control or formats. The changes to be described below were either found to be generally useful or were made to remedy failures of the subroutine in particular applications. All symbolic cards which have been modified are labelled MOD 1 to 20; those labelled SCHR0004 to 176 are unaltered from the original SHARE distribution.

MOD 2 - The Call Statement

The variables which must be supplied are

NI,NS: Output control parameters whose function is explained in the comment cards SCHR0004-007.

RMAX,RMIN: Specify the range of r. RMIN $\le$ r $\le$ RMAX.

V,S: Singly-dimensional arrays containing respectively the numerical potential and the solution, on exit. The source program must contain these in a DIMENSION statement.

N: The number of equally-spaced intervals used in the integration. The length of each interval is h=(RMAX-RMIN)/N.

KV: the number of nodes in the solution. See comments on this below.

E0: Upon entry, E0 is the first trial eigenvalue provided to SCHR by the user's program. E0 must be a negative number. At exit time it is replaced by the E calculated on the last iteration. (Note: the second symbol in this variable name is a zero, not the letter O.)

EPS: The convergence criterion is $E' - E \leq EPS$ where $E' = E + \Delta E$. One may use EPS = 0., except in rare circumstances where the magnitude of the solution E is extremely small relative to the V_i's.

MAXIT: Is the maximum number of the iterations to be performed. It might be noted that frequently the iteration procedure will not satisfy a zero epsilon test. For eigenfunctions with an odd number of nodes especially it can oscillate between two values differing by only 1 unit in the last figure. Hence, considerable time can be wasted by making MAXIT unnecessarily large. In most cases this could be prevented by making $\epsilon = 2x10^{-8}$ E_{trial}. A convenient way of doing this would be to insert the following statement near the beginning of the subroutine, following MOD 2, say,

```
EPS = 2.0 - 8 * EO
```

Cooley has written this subroutine as a FUNCTION with the following purpose in view. At exit time, SCHR = 0. if convergence has been achieved in MAXIT iterations or less. Otherwise, SCHR = 1. and EO and S are the results for MAXIT iterations. Therefore, if one wants to test for convergence one should write in the source program

```
IF [SCHR(...)] N1, N2, N3.
```

When not testing for convergence, one may write

```
CALL SCHR (...).
```

MOD 3 - The Range of Integration

Cooley's published application of his program was to the H_2^+ molecule-ion. The values of its eigenfunctions at $r = 0$ are in the range of 10^{-25}. For most molecules the values at $r = 0$ are much smaller than 10^{-39}, which produces a machine zero on the 7090. Hence the use of $r = 0$ as a lower bound needlessly extends the range of the integration. Values of RMIN = r_e-2 and RMAX = r_e+5 (in atomic units) will give a satisfactory range for molecules having $\omega_e/\omega_e x_e \simeq 50$. The larger this ratio, the more the range may be reduced. It is well to print the eigenvalues at the two extremes in order to be sure that they are sufficiently small.

MOD 4-6 - Trial Eigenvalue too High

It can happen that an iteration will result in a large correction to the trial eigenvalue and place the new trial eigenvalue beyond the range of bound states. Cooley's provision for this was to replace such a trial value by the third last potential value, V(N-2), and then continue with the next iteration. Our experience was that when such a failure occurred resumption of the iteration through this artifice never led to the eigenvalue which was being sought originally. Hence we preferred to terminate the procedure with a print-out to indicate the source of the failure.

MOD 7-11 - Test for Crossing Point

Cooley used three separate tests for terminating the inward integration: a decrease in the eigenfunction, an increase in the potential, and, if neither of these tests did it before the value r_2 was reached, it was made the crossing-point. We have removed the test on the potential for the following reason. In deriving some RKR-type potentials various interpolation schemes were used. In some cases it was found that small local irregularities could be introduced, especially near the classical turning points or near the extremities of the potential where various approximate forms were joined on to the RKR segment. If the inward integration is terminated very far away from the maximum of the eigenfunction, the efficiency of the iteration procedure is greatly reduced and may result in failure of the program. It is necessary to retain the IF(M-2) test, for if the eigenfunction test were never satisfied a tight loop would result. The number of cases in which termination was effected by this test were very few, and all were associated with potentials generated in part by interpolation. The detailed reasons for failure in these cases were never ascertained but were thought to be closely related to the small irregularities arising from unsatisfactory interpolation procedures. The print-out of the crossing-point (MOD 8-11) was made principally to investigate the difficulties encountered in these instances.

MOD 12-15 - Node-Count

Frequently the normalized solutions will have values smaller in magnitude than 10^{-39} near either end of the integration. On the 7090 underflow will occur producing a zero, but one which

The modifications made here prevent an oscillation in the signs of these zeros from being counted as a node. In connection with the node-count we might point out a danger in specifying by a number rather than by a variable any parameter in a CALL statement. If the parameter is changed by the subroutine its new value will be used in any subsequent Fortran statements employing that parameter. If one used the statement CALL SCHR (...,...,5,...) where 5 specified KV, errors could easily result if the node-count actually differed from 5. For instance, if 6 nodes were counted a later program statement J = J+5 would be executed as J = J+6. (A number used for E0 in the CALL will be altered on exit.) In this program the input KV is never used so there is no point to specifying it numerically anyway.

MOD 16-19 - Print Control

This modification is included merely as an example of an additional output option controlled by the parameter NS. With NS = 10 and NI > 9 only the node-count and eigenvalues are printed. It should be noted that MOD 17 and MOD 4 are not under the control of NI, nor have we included the option of on-line printing for these output statements. While it is most unlikely that a 7090 user would ever be printing results on line, the option provided by Cooley has not been removed from this deck. It is useful since some monitor systems interpret the PRINT command as WRITE OUTPUT TAPE 3. Where this is the case the user would have to replace cards MOD 4 and 17 by corresponding PRINT statements, but could use Cooley's output formats by setting the appropriate control parameter equal to zero.

A Numerical Example

For the convenience of anyone wishing to check out a program which uses SCHR, we provide the following test case. Using the input parameters N = 1000, RMAX = 7.5 and RMIN = 0.5 generate an array of r values in which

$$R(1) = RMIN + H$$

where

$$H = (RMAX - RMIN)/N$$

and

$$R(J) = R(J - 1) + H, \qquad J = 2,N$$

Use these values to generate the Morse potential, Eq. (4), with the parameters $D_e = 605.559$, $r_e = 2.40873$ and $\beta = 0.988879$. [Note that the value of V(RMIN) is not used. V(1) = V(RMIN + H).] Call SCHR after specifying the trial eigenvalue E0 = -581.46902. On exit print out the values of R(300), V(300), S(300) and E0. A program intended for generating vibrational-rotational eigenfunctions will include provision for altering the rotationless potential to the one appropriate for any J state. Use it to provide the V array for J = 20, change E0 to -508.62023, call SCHR and then make the same print-outs as before. The values which should be obtained are given in Table I.

The first trial eigenvalues given in the text above were calculated from equations which may be found in reference 2. For purposes of this test there should be no need to specify them beyond two figures unless the value of MAXIT is very small.

Table 1. Morse Eigenvalues for the J=0 and J=20 rotational states of HCl. Values of the potentials and the eigenfunctions at r=2.5999998 atomic units are also given.

	J=0	J=20
E	-581.46913	-511.65467
V(r)	-587.57499	-525.44473
S(r)	1.1936730	1.4065177

For the sake of completeness the potentials and eigenfunctions obtained in the test are shown in Figures 1 and 2.

Finally, it might be noted that appendices A and B are photographic reproductions of the machine listings of the two decks. These decks were compiled and employed in a test program which used every statement with the acception of the on-line print commands. No errors were detected.

References

1. J. W. Cooley, Math. of Computation, XV, 363 (1961).
2. J. K. Cashion, "The Testing of Diatomic Potential Energy Functions by Numerical Methods," UCRL 10643, May 1963 and J. Chem. Phys., in press.
3. J. K. Cashion, "Vibration-Rotation Interaction Factors for Diatomic Molecules Calculated by Numerical Methods," UCRL-10644.
4. R. N. Zare, to be published.
5. G. Herzberg, "Spectra of Diatomic Molecules," 2nd Ed., D. Van Nostrand, New York, 1950.

Captions for Figures

1. Morse potentials for HCl with J = 0 and J = 20.

2. Ground state eigenfunctions for the potentials shown in Fig. 1.



300
200
100
0
U(r)
-100
-200
-300
-400
-500
-600
2
3
4
5
r, atomic units
J = 0
J = 20
MU-30928

Fig. 1.



2.0
1.5
1.0
0.5
0
2.0
2.5
3.0
r, atomic units
ψ0,0
ψ0,20
MU-30929

Fig. 2.

APPENDIX A

```
       REM NU EFM,EFM AND LFM FOR FORTRAN 2                              EFMT  02
       REM                                                               EFMT  03
       ENTRY EFM
       ENTRY EFMT
LFM    OCT 476000000004                                                  EFMT  16
       TTR 1,4                                                           EFMT  17
EFMT   CLA 1,4                                                           EFMT  18
       STA SETN+1                                                        EFMT  19
       CLA TTR                                                           EFMT  20
       TXI EFM+1,4,-1                                                    EFMT  21
EFM    CLA HPR                                                           EFMT  22
       STO OVER+2                                                        EFMT  23
       OCT 476000000002                                                  EFMT  24
SET    CLA X                        TTR  ANALY                           EFMT  25
       STO 8                                                             EFMT  26
       TTR 1,4                                                           EFMT  27
X      TTR ANALY                                                         EFMT  28
ANALY  SXD SAVEX,1                                                       EFMT  29
       LXD 0,1                                                           EFMT  30
       TTR *+14,1                                                        EFMT  31
       TTR MQO                     MQ OVER                               EFMT  32
SAVEX  PZE                          SAVE  XA IN DEC.,THEN  ACC.          EFMT  33
       TTR ACMQ                     11                                   EFMT  34
       TTR AC                        10                                  EFMT  35
       TTR MQ                        9                                   EFMT  36
ZERO   PZE                                                               EFMT  37
       TTR ACMQO                   AC AND MQ OVER                        EFMT  38
       TTR ACO                                                           EFMT  39
AC     CLA ZERO                                                          EFMT  40
       TTR RETRN                                                         EFMT  41
ACMQ   CLA ZERO                     3                                    EFMT  42
       TTR MQ                                                            EFMT  43
MQ     LDQ ZERO                     1, MQ UNDER                          EFMT  44
RETRN  LXD SAVEX,1                                                       EFMT  45
       STO SAVEX                                                         EFMT  46
       CLA 0                                                             EFMT  47
       STA *+2                                                           EFMT  48
       CLA SAVEX                                                         EFMT  49
       TTR **                                                            EFMT  50
ACMQO  ORA MAX                                                           EFMT  51
MQO    STO SAVEA                                                         EFMT  52
       LLS 0                                                             EFMT  53
       LDQ MAX                                                           EFMT  54
       LRS 0                                                             EFMT  55
       TTR OVER+1                                                        EFMT  56
ACO    ORA MAX                                                           EFMT  57
OVER   STO SAVEA                                                         EFMT  58
       CLA 0                                                             EFMT  59
       HPR 63                      PAUSE OVERFLOW                        EFMT  60
       CLA SAVEA                   PRESS START TO CONTINUE               EFMT  61
       TTR RETRN                                                         EFMT  62
SAVEA  PZE                                                               EFMT  63
THREE  PZE 0,0,3                                                         EFMT  64
TTR    TTR SETN                                                          EFMT  65
HPR    HPR 63                                                            EFMT  66
```

```
SETN   ANA THREE
       STO -
       TTR OVER+3
MAX    OCT 377777777777
       END
```

APPENDIX B

```
C       MODIFIED SYMBOLIC DECK FOR COOLEYS D2 NU SCHR 1072                MOD   1
        FUNCTION SCHR(NI,NS,RMAX,RMIN,V,N,KV,E0,EPS,MAXIT,S)              MOD   2
C       NI=0 PRINT ITER. ON-LINE                                          SCHR0004
C       NI=1,..,9  PRINT OFF-LINE TAPE NI                                 SCHR0005
C       NI=OTHERWISE, DONT PRINT                                          SCHR0006
C       NS, SAME EXCEPT FOR SOLUTIONS                                     SCHR0007
        DIMENSION S(500),V(500),P(500),Y(3)                               SCHR0008
C                                                                         SCHR0009
        IF(NI)6,2,1                                                       SCHR0010
 1      IF(NI-10)4,6,6                                                    SCHR0011
 2      PRINT 3                                                           SCHR0012
        GO TO 6                                                           SCHR0013
 3      FORMAT(70H0 ITER          E                F(E)            DF(E)  SCHR0014
     X       D(E)       )                                                 SCHR0015
 4      WRITE OUTPUT TAPE NI,3                                            SCHR0016
 6      CALL EFMT(K)                                                      SCHR0017
        H = (RMAX-RMIN)/FLOATF(N)                                         MOD   3
        H2=H**2                                                           SCHR0019
 8      HV=H2/12.                                                         SCHR0020
        E=E0                                                              SCHR0021
        TEST = -1.                                                        SCHR0022
        DE=0.                                                             SCHR0023
C                                                                         SCHR0024
C       START ITER LOOP                                                   SCHR0025
C                                                                         SCHR0026
 12     DO 171 IT=1,MAXIT                                                 SCHR0027
C       .....START INWARD INTEGRATION                                     SCHR0028
 30     P(N)=1.E-30                                                       SCHR0029
 32     GN=V(N)-E                                                         SCHR0030
 34     GI=V(N-1)-E                                                       SCHR0031
C       .....TEST IF E TOO HIGH                                           SCHR0032
        IF(GI) 35, 36, 36                                                 SCHR0033
  35    WRITE OUTPUT TAPE 3,  899                                         MOD   4
 899    FORMAT(36H DIFFERENCE EQUATION TECHNIQUE FAILS   MOD-5 SCHR=1-    MOD   5
        RETURN                                                            MOD   6
 36     P(N-1)=P(N)*EXPF(RMAX*SQRTF(GN)-(RMAX-H)*SQRTF(GI))               SCHR0036
 38     Y=(1.-HV*GN)*P(N)                                                 SCHR0037
 40     Y(2)=(1.-HV*GI)*P(N-1)                                            SCHR0038
C       .....INTEGRATE                                                    SCHR0039
        K=0                                                               SCHR0040
 44     M=N-2                                                             SCHR0041
 46     Y(3)=Y(2)+((Y(2)-Y)+H2*GI*P(M+1))                                 SCHR0042
 48     GI=V(M)-E                                                         SCHR0043
 50     P(M)=Y(3)/(1.-HV*GI)                                              SCHR0044
C       .....TEST FOR OVERFLOW                                            SCHR0045
 52     IF(K)54,70,54                                                     SCHR0046
C       .....OVERFLOW                                                     SCHR0047
 54     K=0                                                               SCHR0048
        M1=M+1                                                            SCHR0049
        PM=P(M1)                                                          SCHR0050
 55     DO 56 J=M1,N                                                      SCHR0051
 56     P(J)=P(J)/PM                                                      SCHR0052
 58     Y=Y/PM                                                            SCHR0053
 60     Y(2)=Y(2)/PM                                                      SCHR0054
 62     Y(3)=Y(3)/PM                                                      SCHR0055
        GI= V(M+1) - E                                                    SCHR0056
```

```
      GO TO 46                                                          SCHR0057
C     .....TEST FOR CROSSING PT.                                        SCHR0058
 70   IF(ABSF(P(M))-ABSF(P(M+1))    ) 90, 90, 72                        SCHR0059
   72 IF(M-2)90,90,81                                                   MOD  7
 81   Y= Y(2)                                                           SCHR0062
 82   Y(2)=Y(3)                                                         SCHR0063
 84   M=M-1                                                             SCHR0064
 86   GO TO 46                                                          SCHR0065
C                                                                       SCHR0066
 90   PM=P(M)                                                           SCHR0067
      MSAVE = M                                                         MOD  8
 92   YIN=Y(2)/PM                                                       SCHR0068
 94   DO 96 J=M,N                                                       SCHR0069
 96   P(J)=P(J)/PM                                                      SCHR0070
C                                                                       SCHR0071
C     .....START OUTWARD INTEGRATION                                    SCHR0072
C                                                                       SCHR0073
 100  P(1)=1.E-20                                                       SCHR0074
 102  Y=0.                                                              SCHR0075
 104  GI=V-E                                                            SCHR0076
 106  Y(2)=(1.-HV*GI)*P                                                 SCHR0077
      K = 0                                                             SCHR0078
 108  DO 132 I=2,M                                                      SCHR0079
 110  Y(3)=Y(2)+((Y(2)-Y)+H2*GI*P(I-1))                                 SCHR0080
 112  GI=V(I)-E                                                         SCHR0081
 114  P(I)=Y(3)/(1.-HV*GI)                                              SCHR0082
C     .....TEST FOR OVERFLOW                                            SCHR0083
 116  IF(K)118,130,118                                                  SCHR0084
 118  K=0                                                               SCHR0085
      I1=I-1                                                            SCHR0086
      PM=P(I1)                                                          SCHR0087
      DO 120 J=1,I1                                                     SCHR0088
 120  P(J)=P(J)/PM                                                      SCHR0089
 122  Y=Y/PM                                                            SCHR0090
 124  Y(2)=Y(2)/PM                                                      SCHR0091
 126  Y(3)=Y(3)/PM                                                      SCHR0092
      GI=V(I1)-E                                                        SCHR0093
      GO TO 110                                                         SCHR0094
C                                                                       SCHR0095
 130  Y=Y(2)                                                            SCHR0096
 132  Y(2)=Y(3)                                                         SCHR0097
C                                                                       SCHR0098
C     .....FINISHED OUTWARD INTEGRATION                                 SCHR0099
 134  PM=P(M)                                                           SCHR0100
      IF(PM)135,149,135                                                 SCHR0101
 135  YOUT=Y/PM                                                         SCHR0102
 136  YM=Y(3)/PM                                                        SCHR0103
 138  DO 140 J=1,M                                                      SCHR0104
 140  P(J)=P(J)/PM                                                      SCHR0105
C                                                                       SCHR0106
C     .....CORRECTION                                                   SCHR0107
C                                                                       SCHR0108
 142  DF=0.                                                             SCHR0109
 144  DO 146 J=1,N                                                      SCHR0110
 146  DF=DF-P(J)**2                                                     SCHR0111
 148  F=(-YOUT-YIN+2.*YM)/H2+(V(M)-E)                                   SCHR0112
      DOLD=DE                                                           SCHR0113
      IF(K)149,150,149                                                  SCHR0114
 149  F=9.999999E+29                                                    SCHR0115
      DF=-F                                                             SCHR0116
```

```
      DE=ABSF(.0001*E)                                                  SCHR0117
      GO TO 152                                                         SCHR0118
  150 DE=-F/DF                                                          SCHR0119
  152 IF(NI)164,158,154                                                 SCHR0120
  154 IF(NI-10)162,164,164                                              SCHR0121
  156 FORMAT( 1H0  I4,2X 1P4E16.7, 5X 31H THE CROSSING POINT OCCURS AT  MOD   9
     1        I4)                                                       MOD  10
  158 PRINT 156,IT,E,F,DF,DE                                            SCHR0123
  160 GO TO 164                                                         SCHR0124
  162 WRITE OUTPUT TAPE NI, 156, IT,E,F,DF,DE,MSAVE                     MOD  11
  164 EOLD = E                                                          SCHR0126
      E=E+DE                                                            SCHR0127
  166 TEST=MAX1F(ABSF(DOLD)-ABSF(DE),TEST)                              SCHR0128
  168 IF(TEST)171,170,170                                               SCHR0129
  170 IF( ABSF( E-EOLD)     -  EPS )  172,172,171                       SCHR0130
  171 CONTINUF                                                          SCHR0131
      SCHR=1.                                                           SCHR0132
      GO TO 173                                                         SCHR0133
C     .....CONVERGED-COUNT NODES                                        SCHR0134
  172 SCHR=0.                                                           SCHR0135
  173 KV=0                                                              SCHR0136
      NL=N-2                                                            SCHR0137
  174 DO 192 J=3,NL                                                     SCHR0138
  176 IF(P(J))178,177,177                                               SCHR0139
  177 IF(P(J-1))180,192,192                                             SCHR0140
  178 IF(P(J-1))192,270,184                                             MOD  12
C     POS. NODE                                                         SCHR0142
  180 IF(P(J+1))192,182,182                                             SCHR0143
  182 IF(P(J-2))190,192,192                                             SCHR0144
C     NEG. NODE                                                         SCHR0145
  184 IF(P(J+1))186,192,192                                             SCHR0146
  186 IF(P(J-2))192,190,190                                             SCHR0147
C     IGNORE FALSE NODE DUE TO UNDERFLOW                                MOD  13
  270 IF(P(J+1))280,192,192                                             MOD  14
  280 IF(P(J-2))192,192,190                                             MOD  15
  190 KV=KV+1                                                           SCHR0148
  192 CONTINUE                                                          SCHR0149
C     .....NORMALIZE                                                    SCHR0150
  200 SN=SQRTF(-H*DF)                                                   SCHR0151
  202 DO 204 J=1,N                                                      SCHR0152
  204 S(J)=P(J)/SN                                                      SCHR0153
C     .....PRINT SOLUTION                                               SCHR0154
  206 IF(NS)236,210,208                                                 SCHR0155
  208 IF(NS-10)210,236,236                                              SCHR0156
  210 DO 234 JF=1,N,300                                                 SCHR0157
  212 IF(NS)236,216,218                                                 SCHR0158
  214 FORMAT(47H1SCHR- SOLUTION OF RADIAL SCHR. EQUATION FOR V= I3, 7H  SCHR0159
     X   E= 1PE15.7 /20H0   I      S(I)    5(20H    I        S(I)     ))SCHR0160
  216 PRINT 214,KV,E                                                    SCHR0161
      GO TO 220                                                         SCHR0162
  218 WRITE OUTPUT TAPE NS,214,KV,E                                     SCHR0163
  220 JL=XMINOF(JF+49,N)                                                SCHR0164
  222 DO 234 J=JF,JL                                                    SCHR0165
  224 IL=XMINOF(J+250,N)                                                SCHR0166
  226 IF(NS)236,230,232                                                 SCHR0167
  228 FORMAT(6(I5,1PE15.7))                                             SCHR0168
  230 PRINT 228,(I,S(I),I=J,IL,50)                                      SCHR0169
      GO TO 234                                                         SCHR0170
  232 WRITE OUTPUT TAPE NS,228,(I,S(I),I=J,IL,50)                       SCHR0171
  234 CONTINUE                                                          SCHR0172
```

```
236    E0=E                                                          SCHR0173
       IF(NS-10)250,875,250                                          MOD 16
875    WRITE OUTPUT TAPE 3,  876, KV,E                               MOD 17
876    FORMAT(54H0 SOLUTION OF THE RADIAL SCHRODINGER EQUATION FOR V =     MOD 18
     1 I3,8H    E =   1PE15.7 )                                      MOD 19
250    RETURN                                                        SCHR0174
       FREQUENCY 52(0,1,0),70(0,0,1),72(0,0,1),94(100),55(50),108(100),11 SCHR0175
     X 6(0,1,0),138(100),144(200),152(1,0,0),154(0,1,1),202(200)    SCHR0176
       END                                                           MOD 20
```